
Subject: Re: cGImage, Multiplot with /KEEP_ASPECT_RATIO
Posted by [David Fanning](#) on Tue, 12 Feb 2013 17:06:48 GMT
[View Forum Message](#) <> [Reply to Message](#)

David Fanning writes:

```
> Alright, this was easy enough to do, so I just had Coyote do it. :-)
>
> Find it here:
>
> http://www.idlcoyote.com/programs/cgaspect.pro
>
> And here is how I would use the routine to display these two images with
> colorbars:
>
> ;-----
> img1 = cgDemoData(7)
> img2 = congrid(img1, 360, 180)
> cgDisplay, 900, 450
>
> pos = cgLayout([2,1], OYMargin=[4, 11])
> cgImage, img1, /KEEP_ASPECT_RATIO, CTIndex=3, $
> /AXES, POSITION=pos[*,0], OPOSITION=op
> cgColorbar, CTIndex=3, /Fit
>
> pos2 = pos[*,1]
> ydiff = op[3] - pos2[3]
> pos2[3] = op[3]
> pos2[1] = pos2[1] + ydiff
>
> img2pos = cgAspect(Position=pos2, Aspect=img2, Align='top')
> cgImage, img2, /AXES, POSITION=img2pos, CTIndex=2, /NoErase
> cgColorbar, CTIndex=2, /Fit
> END
> ;-----
```

That's a little convoluted. Just for the record, here is a better formulation that takes better advantage of cgAspect.

```
;----- --
img1 = cgDemoData(7)
img2 = congrid(img1, 360, 180)
cgDisplay, 900, 450

pos = cgLayout([2,1], OYMargin=[4, 11], OXMargin=[5, 8], XGap=6)
pos1 = pos[*,0]
img1pos = cgAspect(Position=pos1, Aspect=img1, Align='top')
cgImage, img1, CTIndex=3, /AXES, POSITION=img1pos, OPOSITION=op
```

```
cgColorbar, CTIndex=3, /Fit
```

```
pos2 = pos[*,1]  
ydiff = op[3] - pos2[3]  
pos2[3] = op[3]  
pos2[1] = pos2[1] + ydiff
```

```
img2pos = cgAspect(Position=pos2, Aspect=img2, Align='top')  
cgImage, img2, /AXES, POSITION=img2pos, CTIndex=2, /NoErase  
cgColorbar, CTIndex=2, /Fit  
END  
;----- --
```

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>

Sepore ma de ni thue. ("Perhaps thou speakest truth.")
